UNCLASSIFIED

AD NUMBER

AD844545

NEW LIMITATION CHANGE

TO

Approved for public release, distribution unlimited

FROM

Distribution authorized to U.S. Gov't. agencies only; Administrative/Operational Use; 26 OCT 1968. Other requests shall be referred to Naval Ordnance Lab., White Oak, MD.

AUTHORITY

USNOL ltr, 12 Dec 1972

THIS PAGE IS UNCLASSIFIED

AD-844545 1

ORD-913

NOLTR 68-138

TR-68-138 C.1

EFFECTS OF SURFACE REFLECTIONS ON SHOCK WAVE IMPULSE

26 OCTOBER 1968

NOL

UNITED STATES NAVAL ORDNANCE LABORATORY, WHITE OAK, MARYLAND

NOLTR 68-138

EFFECTS OF SURFACE REFLECTIONS ON SHOCK WAVE IMPULSE

by

Verna K. Shuler

ABSTRACT: Under free-field conditions, the impulse of an underwater explosion shock wave is constant at a given radial distance from the explosive charge. If either the charge or recording gage is at shallow depth, however, the rarefaction wave reflected back into the water from the interface (the "surface cut-off wave") decreases the duration of the positive shock wave so that the radius of constant impulse is significantly decreased from its free-water value. The effect of surface cut-off is discussed and illustrated in this report. An equation for shock wave "partial impulse," which takes into account the effects of surface cut-off, is discussed and a computer program that generates a locus of points of constant partial impulse is described.

UNDERWATER EXPLOSIONS DIVISION
EXPLOSIONS RESEARCH DEPARTMENT
U. S. NAVAL ORDNANCE LABORATORY
WHITE OAK, SILVER SPRING, MARYLAND

EFFECTS OF SURFACE REFLECTIONS ON SHOCK WAVE IMPULSE

The impulse of an underwater explosion shock wave, a parameter that is often important in determining the damaging effect of the explosion, may be reduced to a small fraction of its free-water value if either the charge or the target is at very shallow depth. This report discusses a semi-empirical equation that can be used to approximate the near-surface shock wave impulse for various test configurations.

An iterative machine program for estimating contours of constant impulse is also presented. This program, which is flexible enough to be used for a number of applications, is a significant improvement over the tedious and time-consuming manual methods previously used for estimating such contours.

This work was performed under ORDTASK No. ORD-332-001/UF17-354-304, Explosions Underwater Research and Technology.

E. F. SCHREITER
Captain, USN
Commander

C. J. Aronson

C. J. ARONSON
By direction

CONTENTS

REFERENCES

(a) A. B. Arons, "Underwater Explosion Shock Wave Parameters at Large Distances from the Charge," J. Acoust. Soc. Am., 26, 343-346, 1954.

(b) A. B. Arons, D. R. Yennie, and T. P. Cotter, Jr., Underwater Explosions Research, Off. Naval Res. Compendium, Vol. I, 1473-1583, (1949).

(c) J. P. Slifko, "Pressure Pulse Characteristics of Deep Explosions as Functions of Depth and Range," NOLTR 67-87, Sep 1967.

ILLUSTRATIONS

EFFECTS OF SURFACE REFLECTIONS ON SHOCK WAVE IMPULSE

1. INTRODUCTION

From various underwater explosion tests conducted in the past, empirical relationships representing the shock wave impulse for particular situations have been developed. Arons (references a and b)* describes the impulse at various ranges from a shallow charge in free water as a function of charge weight and range. His empirical power functions were derived from pressure-time records integrated over relatively long times, to a point where the pressure had decayed to low amplitude. Slifko (reference c) has developed an empirical relationship for the impulse of the entire first positive phase of pressure waves measured at long ranges from deep charges in free water.

A situation frequently of interest in practical applications, for which no provision is made in either of the above-mentioned relationships, is one in which surface reflections arrive at the gage shortly after the shock wave arrival. When the negative pressure in the surface-reflected wave interacts with the positive direct pressure wave, it cancels, or "cuts-off," the pressure in the tail of the shock wave, thereby decreasing the impulse at the gage location.

Because surface reflections significantly reduce the shock wave impulse when either the charge or the gage is relatively shallow, a semi-empirical equation for shock wave impulse that includes the effects of surface reflections has been developed and is discussed in this report. This quantity is called "partial" impulse to differentiate it from the free water relationship given by Arons. Like that of Arons, it is a function of charge weight and range, but the time to which the pulse is integrated is included as a variable in the equation.

In addition to the discussion of the partial impulse equation, the report includes a computer program which computes the locus of points at which the underwater shock wave impulse is constant. Instructions for using the program are also included.

2. SEMI-EMPIRICAL EXPRESSION FOR SHOCK WAVE IMPULSE

2.1 Assumed Wave Form

The pressure in the shock wave from an underwater explosion rises almost instantaneously to a maximum and then decays to ambient pressure at a rate that depends on the weight of the explosive charge and the distance the wave has travelled outward. The initial rate of decay can be represented quite accurately by an exponential; at later times, however, the pressure decays more slowly so that the approximation of the wave form by the initial exponential underestimates the pressure in the tail of the shock wave. Since this report is concerned primarily with situations in which the surface cut-off occurs soon after the shock wave arrival, the error introduced by using the exponential approximation is negligible for many conditions of practical interest. The pressure wave for which impulse is computed in this report is assumed to be an exponential of the form:

---

* References are given on page iii.

$$P(t) = P_m \exp(-t/\theta) \qquad (1)$$

where the peak pressure, P_m, and time decay constant, θ, conform to the TNT similitude values given by Arons (reference a):

$$P_m \text{ (psi)} = 2.16 \times 10^4 \, (W^{1/3}/R)^{1.13} \qquad (2)$$

$$\theta \text{ (msec)} = 6 \times 10^{-2} \, W^{1/3} \, (W^{1/3}/R)^{-0.22} \qquad (3)$$

where: W = charge weight (lb)
R = slant range (ft).

2.2 Partial Impulse

The shock wave impulse, I, is:

$$I = \int_0^{t_1} p \, dt \qquad (4)$$

and from Equations 1 and 4, the partial impulse, I_p, as a function of the integration time, is:

$$I_p = P_m \, \theta \left[1 - \exp(t_1/\theta)\right] \qquad (5)$$

and from Equations (2) and (3):

$$I_p \text{ (psi-msec)} = 1300 \, W^{0.64} \, R^{-0.91} \left[1 - \exp(Q)\right] \qquad (6)$$

where: $Q = -16.67 \, t_1 W^{-0.26} \, R^{-0.22}$.

Integration time, t_1, for the partial impulse may be either some specified constant time, t_o, or surface cut-off time, t_s. Since the surface reflected wave of negative amplitude cancels the positive pressure in the shock wave as shown in Figure 1, maximum possible integration time is surface cut-off time.

2.3 Surface Cut-Off Time

Assuming regular acoustic reflection and constant wave propagation velocity, surface cut-off time, t_s, is determined by the particular charge-gage geometry. From the geometric relationships shown in Figure 2, the following equation for surface cut-off time can be derived if the velocity of sound in the water is assumed to be 5 ft/msec:

$$t_s \text{ (msec)} = 0.2 \left[(R^2 + 4yD)^{1/2} - R \right] \tag{7}$$

where: R = slant range from charge to gage, ft
Y = gage depth, ft
D = charge depth, ft.

An equation has been derived for the locus of gage positions at which surface cut-off time is constant. Such a locus is the lower branch of a hyperbola with a vertical transverse axis through the charge. The general equation for the hyperbola is:

$$\frac{Y^2}{a^2} - \frac{X^2}{b^2} = 1 \tag{8}$$

where: $a = 1/2\ \Delta R$
$b = 1/2\ (4D^2 - \Delta R^2)^{1/2}$
X = horizontal range
ΔR = difference in path length of the reflected wave and the path length of the direct wave (also $\Delta R = t_s V$, where V = velocity of sound in the water).

Figure 3 shows several contours of constant surface cut-off time for a charge at 865-ft depth. At the water surface $t_s = 0$; on each narrowing hyperbolic branch t_s increases and finally reaches a maximum value of 0.346 sec when the gage is directly below the charge. Since t_s is determined from purely geometric considerations, the charge and gage locations may be used interchangeably. In other words, if the gage, rather than the charge, is held at a constant depth of 865 feet, Figure 3 shows how surface cut-off time varies as charge location changes.

2.4 Effect of Surface Cut-Off Time on Partial Impulse

Figures 4 and 5 illustrate the effect of surface cut-off on shock wave impulse at a given horizontal distance from the charge. In these figures, 10-lb charges are located at 2-ft and 20-ft depths, respectively, and horizontal range is held at constant 20 feet. Maximum pressure (Equation 2) and partial impulse (Equation 6), each normalized to its value at 1-ft depth, is plotted versus gage depth; integration time is surface cut-off time.

In Figure 4, as the gage is lowered from the surface down to about 8 feet, peak pressure changes very little because slant range varies only slightly. Partial impulse, however, increases significantly with increasing gage depth until it reaches a maximum of about 4.5 times its 1-ft value when the gage is about 13 feet deep. Increasing surface cut-off time is responsible for the increase in impulse until the gage reaches about 13 feet; at greater depths, slant range increases enough to offset the surface cut-off effects and impulse decreases sharply. The pressure also drops off more rapidly for the greater gage depths because of the increasing slant range.

With the deeper charge (20 feet) of Figure 5, there is less variation in shock wave impulse as the gage is lowered. The impulse here varies in the same fashion as peak pressure; both parameters are at a maximum when the charge and gage are at the same 20-ft depth. Impulse is not greatly affected by surface cut-off when the charge is at 20 feet because surface reflections arrive relatively late.

3. CONTOURS OF CONSTANT IMPULSE

3.1 Modification of Free-Field Contour

Under free-field conditions, the contours of constant impulse lie along spherical surfaces centered at the explosion. For a given weight of charge, the radius at which the free-field impulse has some particular constant value can be found from Arons' empirical equation for shock waves integrated to 6.7 θ (reference b):

$$I\ (\text{psi-msec}) = 1780\ W^{1/3}\ (W^{1/3}/R)^{0.94} \tag{9}$$

Near the water surface, however, the radius must be found from a partial impulse function such as Equation 6. At some point, Equation 6 becomes a poor approximation, as discussed below, and empirical corrections are needed to relate the two impulse functions. Unfortunately, only fragmentary data obtained at relatively short ranges are available so that it is not possible to empirically define contours for all conditions of interest. A sample contour that could be estimated from available data is shown in Figure 6.

In Figure 6, the heavy line is the estimated contour along which I = 100 psi-msec for a 10-lb charge detonated at 10 feet. For shallow depths, this contour coincides with the partial impulse contour computed from Equation 6 of this report; at greater depths the contour coincides with the free-field contour of Arons. These two limiting curves are shown dashed beyond the point where they coincide with the desired contour. Interpolation between the two limiting contours was based on the fragmentary data mentioned above.

3.2 Comparison of Contours

It can be seen from Figure 6 that for these particular conditions, the partial impulse equation discussed in previous sections of this report provides a poorer fit as the gage location deepens, and gives an estimated slant range that is about 15% too small when the gage depth is 40 feet. For a different set of conditions, the partial impulse equation given here might be an excellent approximation down to much greater depths, but there is no simple, straightforward way to extrapolate Figure 6 to other conditions. The adequacy of the exponential approximation must be reckoned in terms of the number of time constants (θ of Equation 3) over which the wave is integrated, and there is no simple generalized expression for this relative quantity, $t = t_1/\theta$. The time constant, θ, depends upon only charge weight and range, while t depends upon charge weight, total geometry, level of constant impulse of interest and upon the maximum integration time if such a limit is imposed. We do not have adequate data to define empirical corrections to Equation 6 for all cases of interest; however, certain generalizations can be made. If t_1 is no greater than about 3 θ, Equation 6 probably gives an adequate approximation of I_p. In terms of the variables noted on Figure 6, this means that the exponential approximation will fit the true contour down to deeper gage depths if (a) charge weight is increased

or (b) the impulse level of the desired contour is decreased or (c) the charge is fired at a shallower depth. When sufficient data are available, generalized methods of generating the contours of constant impulse for a range of variables will be developed.

4. COMPUTER PROGRAM FOR ESTIMATING LOCI OF CONSTANT PARTIAL IMPULSE

4.1 General

Since the range corresponding to a constant partial impulse cannot be found explicitly from Equation 6, estimating loci of constant partial impulse by manual computation is tedious and time-consuming. With the iterative computer program outlined below, however, the range at which I_p is within ± 0.1 psi-msec of some constant value, I_o, can be found rapidly.

4.2 Outline of Program

The program has been written in the Basic Language on the remote computing facility at NOL. It finds the maximum range to which a charge of weight W, at depth D, can deliver some given impulse I_o, within some specified time t_o. Although t_o is introduced as a constant value in the program listing shown here, it can be introduced in functional form with only minor program changes. Note that the maximum possible duration of the shock wave is the surface cut-off time, t_s, of Equation 7; consequently, the program compares the input parameter t_o with t_s for the geometry in question and uses t_s rather than t_o in computing the partial impulse whenever $t_o > t_s$. If the desired integration time is always t_s, then a large value for t_o (such as 10 sec) should be inserted in the appropriate data statement.

A detailed explanation of the program logic is not included in the report because sufficient comments are scattered throughout the program so that one who understands the programming language should be able to follow the logic. The input parameters from which the computer completes all computations are as follows:

(1) charge weight, W, lb
(2) charge depth, D, ft
(3) gage depth, Y, ft
(4) given impulse, I_o, psi-msec
(5) given integration time, t_o, msec.

The iteration involves these four basic steps:

(1) A slant range, R, is automatically selected and surface cut-off time, t_s, for the particular geometric situation is calculated.
(2) Partial impulse, I_p, is computed using this same slant range, R, gage depth, Y, and charge weight, W. Integration time, t, is the smaller of t_o or t_s.
(3) Partial impulse from step (2) is compared with the given impulse, I_o; if $I_p < (I_o - 0.1)$, then range is automatically decreased or if $I_p > (I_o + 0.1)$, range is automatically increased. The new range is R' and corresponding surface cut-off time is t_s'.
(4) The process of computing I_p and comparing with I_o is repeated until a range at which $I_p = I_o \pm 0.1$ psi-msec is found.

If, for some reason, there is no convergence after 50 iterations, a message indicating where adjustments should be made will be printed. If the conditions are such that

there is no solution, this also will be indicated. A list of the program and samples of output are included in Appendices A and B to this report.

4.3 Data Arrangement

All data for the Basic program is entered in data statements between statements 5000 and 7000. The statement numbers used here as examples are chosen for illustration purposes; the user may number his data statements as he chooses:

5001 Data for this statement should be:
- (a) number of charge weights
- (b) number of gage depths
- (c) charge depth (ft).

5002 This data statement contains the individual charge weights (lb) given--there may be 20 of these values; the number of charge weights is indicated in 5001 (a).

5003 Each individual gage depth (ft) is listed in this statement--again, there may be 20 gage depths given. The number of gage depths given is specified in 5001 (b).

5004 In this statement, the desired impulse (I_o, psi-msec) and integration time (t_o, msec) are given, respectively.

The data arrangement is illustrated in the following example: For 10-lb and 100-lb charges fired at 40-ft depth, determine the ranges at which I_o = 150 psi-msec when t_o = 2.5 msec. Use gage depths of 10, 20, and 50 feet. Appropriate data statements follow:

```
5001  Data 2, 3, 40
5002  Data 10, 100
5003  Data 10, 20, 50
5004  Data 150, 2.5
```

Note that a new run is required if charge depth, given impulse, or given integration time is changed.

ACKNOWLEDGMENT

The author is indebted to Miss E. A. Christian of the Explosions Research Department at NOL for her direction in this work.



POSITIVE WAVE "CUT-OFF" AT TIME t_s

FIG. 1 SURFACE CUT-OFF OF UNDERWATER SHOCK WAVE



FIG. 2 UNDERWATER GEOMETRY



FIG. 3 CONTOURS OF CONSTANT SURFACE CUT-OFF TIME, t_s, FOR CHARGE DEPTH OF 865 FEET



FIG. 4 RELATIVE VALUES OF SHOCK PARAMETERS VS GAGE DEPTH WHEN CHARGE IS AT 2-FT DEPTH



FIG. 5 RELATIVE VALUES OF SHOCK PARAMETERS VS GAGE DEPTH WHEN CHARGE IS AT 20-FT DEPTH



FIG. 6 CONTOURS OF CONSTANT IMPULSE

APPENDIX A

LISTING OF COMPUTER PROGRAM

```
10 DIM R(50), I(50), T(50), Y(20), W(20)
50 READ C,G,D
55 REM  C=NUMBER ØF CHG WTS   G=NUMBER ØF GAGE DPHS   D=CHG DPH
60 FØR L = 1 TØ C
70 READ W(L)
75 REM W = INDIVIDUAL CHARGE WEIGHTS
80 NEXT L
90 FØR M = 1 TØ G
100 READ Y(M)
105 REM Y = INDIVIDUAL GAGE DEPTHS
110 NEXT M
112 PRINT
113 PRINT
115 PRINT "CHARGE DEPTH = " D" FEET"
118 READ IO,TO
119 REM IO = GIVEN IMPULSE   TO = GIVEN INTEGRATIØN TIME
120 FØR K = 1 TØ C
122 PRINT
124 PRINT
130 PRINT "CHARGE WEIGHT (LB) =" W(K)
140 FØR M = 1 TØ G
141 LET X1 = 1000
143 LET F = 0
144 LET H = 0
145 PRINT
147 PRINT
150 PRINT"GAGE DEPTH (FT) = "Y(M)
190 LET T(M) = TO
195 LET A1 = IO+.1
197 LET B1 = IO-.1
200 PRINT "GIVEN IMPULSE (PSI-MSEC) = " IO
210 PRINT "GIVEN INTEGRATIØN TIME (MSEC) = "T(M)
220 LET J = 1
230 LET R(J) = (4*Y(M)*D - (5*TO)↑2) / (10 * TO)
233 IF R(J) < 0 THEN 400
240 REM RANGE WHERE SURFACE CUT-ØFF TIME = GIVEN INTEG. TIME IS R(J).
250 GØSUB 1000
260 IF I(J) >=IO THEN 500
262 LET X1 = R(J)
265 REM MAX RANGE SØ FAR IS X1.
270 REM CØMPUTED IMPULSE IS LESS THAN GIVEN IMPULSE.
278 LET F = 1
```

```
280 FØR J = 2 TØ 50
290 LET Q = J-1
300 LET R(J) = .5*R(Q)
310 REM TRYING A RANGE 1/2 ØF PREVIØUS RANGE.
320 GØSUB 1000
330 IF I(J) >= I0 THEN 700
350 REM FINDING A RANGE SMALL ENØUGH S 'HAT GIVEN IMP CAN BE DELIVERED.
360 NEXT J
370 PRINT"AFTER 50 ITERATIØNS ØN DECREASED RANGE,GIVEN IMP NØT REACHED"
390 GØ TØ 5000
400 LET H = 1
402 GØSUB 3000
404 IF Y(M) = D THEN 426
406 REM CASE WHERE CHARGE AND GAGE LIE ØN A VERTICAL LINE.
408 LET R(J) = ABS(Y(M) - D)
410 GØSUB 1000
412 IF I(J) >= I0 THEN 420
415 PRINT"SINCE IMPULSE CALCULATED AT THE LARGER ØF GIVEN INTEGRATIØN"
416 PRINT"TIME ØR SURFACE CUT-ØFF TIME IS LESS THAN THE GIVEN DESIRED"
417 PRINT"IMPULSE, THERE IS NØ SØLUTIØN."
418 GØ TØ 995
420 REM SINCE CØMPUTED IMP>GIVEN IMP., RANGE MAY BE INCREASED.
422 LET N1 = R(J)
424 REM MIN RANGE SØ FAR IS N1.
425 GØ TØ 427
426 LET N1 = 0
427 LET X1 = 10
428 IF X1 <=N1 THEN 432
430 GØ TØ 434
432 LET X1 = N1 + 10
434 LET X9 = 1
436 LET X8 = X1
438 FØR J = 2 TØ 50
440 LET R(J) = X9 * X8
442 GØSUB 2000
444 GØSUB 4000
460 GØSUB 1000
470 IF I(J) >= I0 THEN 700
472 LET X8 = X8 / 2
474 NEXT J
480 PRINT"NØ CØNVERGENCE AFTER 50 ITERATIØNS ØN 438."
485 GØ TØ 5000
500 REM SINCE CØMPUTED IMP>GIVEN IMP WHERE SURFACE CUT-ØFF TIME EQUALS
505 REM GIVEN INTEGRATIØN TIME, ATTEMPT IS MADE TØ FIND MAX RANGE AT
510 REM WHICH GIVEN IMP CAN BE DELIVERED WITHIN SURF CUT-ØFF TIME.
534 LET N1 = R(J)
535 REM MIN RANGE IS N1.
540 LET U = 10
545 LET Z3 = R(J)
550 FØR J = 2 TØ 50
570 REM TRYING A RANGE ØF U * Z3.
580 LET R(J) = U*Z3
590 GØSUB 2000
600 GØSUB 1000
```

```
610 IF I(J)>=I0 THEN 700
617 LET X1 = R(J)
619 REM MAX RANGE IS X1.
620 REM CØMPUTED IMPULSE LESS THAN GIVEN IMPULSE.
630 LET U = U/2
640 NEXT J
650 PRINT "NØ CØNVERGENCE AFTER 50 ITERATIØNS ØN PARAG. 550"
660 GØ TØ 5000
700 LET N1 = R(J)
710 REM MIN RANGE SØ FAR IS N1.
712 IF X1 <= N1 THEN 716
714 GØ TØ 720
716 LET X1 = N1 *100
718 REM LETTING MAX RANGE BE X1.
720 FØR J = 3 TØ 50
730 LET R(J) = X1 - (X1-N1)/2
740 REM TRYING A RANGE 1/2 ØF WAY BETWEEN N1 AND X1.
750 IF F = 1 THEN 780
760 GØSUB 2000
770 GØ TØ 790
780 LET T(M) = T0
790 IF H<1 THEN 799
792 GØSUB 4000
799 GØSUB 1000
800 IF I(J) > B1 THEN 825
810 GØ TØ 835
825 IF I(J) < A1 THEN 895
835 IF I(J) > I0 THEN 860
837 REM CØMPUTED IMP NØT WITHIN GIVEN IMP +ØR- 0.1  IMP TØØ SMALL.
845 LET X1 = R(J)
850 REM NEW MAX RANGE IS X1.
855 GØ TØ 880
860 REM CØMPUTED IMP NØT WITHIN GIVEN IMP +ØR- 0.1  IMP TØØ LARGE.
865 LET N1 = R(J)
870 REM NEW MIN RANGE IS N1.
880 NEXT J
885 PRINT "NØ CØNVERGENCE AFTER 50 ITERATIØNS ØN 720."
890 GØ TØ 5000
895 LET K4 = D-Y(M)
896 IF R(J) <ABS(K4) THEN 985
980 IF H = 1 THEN 983
981 IF F = 1 THEN 990
982 GØ TØ 987
983 IF T(M) < T0 THEN 987
984 GØ TØ 990
985 PRINT"NØ SØLUTIØN-----GAGE AND CHARGE ARE TØØ FAR APART."
986 GØ TØ 995
987 PRINT"RANGE ØF"R(J)"FEET HAS BEEN FØUND AT WHICH IMPULSE ØF"I(J)
988 PRINT"CAN BE DELIVERED IN"T(M)"MSEC WHICH IS SURFACE CUT-ØFF TIME."
989 GØ TØ 995
990 PRINT"RANGE ØF"R(J)"FEET HAS BEEN FØUND AT WHICH IMPULSE ØF"I(J)
991 PRINT"CAN BE DELIVERED WITHIN"T(M)"MSEC WHICH IS GIVEN INTEGRATIØN"
992 PRINT "TIME.  SURFACE REFLECTIØNS DØ NØT ARRIVE THIS EARLY."
995 NEXT M
```

```
996 NEXT K
998 GØ TØ 7000
1000 REM THIS SUBRØUTINE CØMPUTES IMPULSE.
1030 LET Q=-16.67*T(M)/W(K)↑.333* (W(K)↑.333/ R(J))↑.22
1040 LET I(J) = 1300*W(K)↑.333*(W(K)↑.333/ R(J))↑.91*(1-EXP(Q))
1090 RETURN
2000 REM THIS SUBRØUTINE CØMPUTES SURFACE CUT-ØFF TIME FØR EACH RANGE.
2030 LET T(M)=((R(J)↑2+4*Y(M)*D)↑.5-R(J)) / 5
2060 RETURN
3000 REM THIS SUBRØUTINE CØMPUTES MAX TS (SURFACE CUT-ØFF TIME).
3001 REM (TO IS GIVEN INTEGRATIØN TIME.)
3002 REM NØ SØLUTIØN FØR TS = TO  MUST FIND MAX TS AND PRØCEED.
3005 IF D < Y(M) THEN 3020
3010 LET T(M) = 2*Y(M) / 5.0
3011 IF D = Y(M) THEN 3017
3015 GØ TØ 3035
3017 RETURN
3020 LET T(M) = 2*D / 5.0
3035 IF T(M) < TO THEN 3045
3040 LET T(M) = TO
3042 REM TO<MAX TS   USE TO IN CØMPUTING IMPULSE.
3043 RETURN
3045 REM MAX TS<TO  USE TS IN CØMPUTING IMPULSE.
3050 RETURN
4000 REM THIS SUBRØUTINE FINDS SMALLER ØF TO AND TS TØ BE USED FØR IMP"
4010 IF TO > T(M) THEN 4050
4020 LET T(M) = TO
4030 REM SINCE TO<TS, TO WILL BE USED IN CALCULATING IMPULSE.
4040 RETURN
4050 REM SINCE TS < TO, TS WILL BE USED IN CALCULATING IMPULSE.
4060 RETURN
5000 PRINT "ITERATIØNS NEED REVISING"
5001 DATA 2 3,40
5002 DATA 10,100
5003 DATA 10,20,50
5004 DATA 150,2.5
7000 STØP
8000 END
```

APPENDIX B

SAMPLE OUTPUT

```
CHARGE DEPTH =  40    FEET

CHARGE WEIGHT [LB] = 10

GAGE DEPTH [FT] =  10
GIVEN IMPULSE [PSI-MSEC] =  150
GIVEN INTEGRATIØN TIME [MSEC] =  2.5
RANGE ØF 53.6331 FEET HAS BEEN FØUND AT WHICH IMPULSE ØF 150.023
CAN BE DELIVERED WITHIN 2.5     MSEC WHICH IS GIVEN INTEGRATIØN
TIME.  SURFACE REFLECTIØNS DØ NØT ARRIVE THIS EARLY.

GAGE DEPTH [FT] =  20
GIVEN IMPULSE [PSI-MSEC] =  150
GIVEN INTEGRATIØN TIME [MSEC] =  2.5
RANGE ØF 53.6223 FEET HAS BEEN FØUND AT WHICH IMPULSE ØF 150.051
CAN BE DELIVERED WITHIN 2.5     MSEC WHICH IS GIVEN INTEGRATIØN
TIME.  SURFACE REFLECTIØNS DØ NØT ARRIVE THIS EARLY.

GAGE DEPTH [FT] =  50
GIVEN IMPULSE [PSI-MSEC] =  150
GIVEN INTEGRATIØN TIME [MSEC] =  2.5
RANGE ØF 53.629  FEET HAS BEEN FØUND AT WHICH IMPULSE ØF 150.034
CAN BE DELIVERED WITHIN 2.5     MSEC WHICH IS GIVEN INTEGRATIØN
TIME.  SURFACE REFLECTIØNS DØ NØT ARRIVE THIS EARLY.

CHARGE WEIGHT [LB] = 100

GAGE DEPTH [FT] =  10
GIVEN IMPULSE [PSI-MSEC] =  150
GIVEN INTEGRATIØN TIME [MSEC] =  2.5
RANGE ØF 190.62  FEET HAS BEEN FØUND AT WHICH IMPULSE ØF 149.909
CAN BE DELIVERED IN .830324 MSEC WHICH IS SURFACE CUT-ØFF TIME.
```

```
GAGE DEPTH [FT] =  20
GIVEN IMPULSE [PSI-MSEC] =  150
GIVEN INTEGRATIØN TIME [MSEC] =  2.5
RANGE ØF 231.402 FEET HAS BEEN FØUND AT WHICH IMPULSE ØF 149.994
CAN BE DELIVERED IN 1.36281 MSEC WHICH IS SURFACE CUT-ØFF TIME.


GAGE DEPTH [FT] =  50
GIVEN IMPULSE [PSI-MSEC] =  150
GIVEN INTEGRATIØN TIME [MSEC] =  2.5
RANGE ØF 261.05  FEET HAS BEEN FØUND AT WHICH IMPULSE ØF 149.955
CAN BE DELIVERED WITHIN 2.5     MSEC WHICH IS GIVEN INTEGRATIØN
TIME.  SURFACE REFLECTIØNS DØ NØT ARRIVE THIS EARLY.
```

DISTRIBUTION LIST

	Copies
Chief of Naval Operations Washington, D. C. 20360 Attn: OP-721	1
Chief of Naval Research Washington, D. C. 20360 Attn: Code 408	1
Code 418	1
Code 466	1
Code 468	1
Commander Naval Ordnance Systems Command Washington, D. C. 20360 Attn: ORD-033	1
ORD-0341	1
ORD-035	1
ORD-05411	1
ORD-0534	1
ORD-9132	1
Commander Naval Ships Systems Command Washington, D. C. 20360 Attn: SHIPS 0354	1
Commanding Officer U. S. Naval Radiological Defense Laboratory San Francisco, California 94135 Attn: Code 222	1
Commanding Officer U. S. .aval Ordnance Station Indian Head, Maryland 20640 Attn: Library Division	1
Lovelace Foundation for Medicine Education and Research 5200 Gibson Boulevard, SE (DASA 01-68-C-0171) Albuquerque, New Mexico 87108 Attn: D. R. Richmond	1
I. G. Bowen	1
Headquarters DASA (STMD) Washington, D. C. 20305 Attn: LTCOL Godden	1
Attn: L. J. Belliveau (SPLN)	1

DISTRIBUTION LIST (continued)

	Copies
Commander Naval Weapons Laboratory Dahlgren, Virginia 22448 Attn: Experimental Officer	1
Commander Naval Weapons Center China Lake, California 93555 Attn: Technical Library	1
Commander Naval Undersea Warfare Center 3202 E. Foothill Boulevard Pasadena, California 91107	1
Director Naval Research Laboratory Washington, D. C. 20390	1
Commanding Officer & Director Naval Ship Research & Development Center Washington, D. C. 20007 Attn: Library Dr. W. W. Murray	 1 1
Commanding Officer & Director Underwater Explosions Research Division Naval Ship Research & Development Center Portsmouth, Virginia 23709	1
Superintendent U. S. Naval Postgraduate School Monterey, California 93940	1
Commander Naval Undersea Warfare Center San Diego Division 271 Catalina Boulevard San Diego, California 92152	1
Commanding Officer U. S. Naval Underwater Weapons Research & Engineering Station Newport, Rhode Island 02840	1
Commander U. S. Naval Weapons Station Yorktown, Virginia 23491 Attn: Research & Development Division	1

DISTRIBUTION LIST (continued)

	Copies
Commanding Officer Naval Torpedo Station Keyport, Washington 98345	1
Commanding Officer U. S. Navy Underwater Sound Laboratory Fort Trumbull New London, Connecticut 06321	1
Commanding Officer U. S. Naval Air Development Center Johnsville, Pennsylvania 18974	1
Commander U. S. Naval Oceanographic Office Washington, D. C. 20390	1
Commanding Officer U. S. Navy Mine Defense Laboratory Panama City, Florida 32402	
Attn: Technical Library	1
Code IUW743	1
Commanding Officer Naval Civil Engineering Laboratory Port Hueneme, California 93041	1
Director Waterways Experiment Station Vicksburg, Mississippi 39181	
Attn: Technical Library	1
G. L. Arbuthnot	1
J. N. Strange	1
Director Ordnance Research Laboratory Pennsylvania State University University Park, Pennsylvania 16802	1
Director Woods Hole Oceanographic Institution Woods Hole, Massachusetts 02543	1
Director Scripps Institute of Oceanography La Jolla, California 92037	1
Defense Documentation Center Cameron Station Alexandria, Virginia 22314	
Attn: TIPDR	20

DISTRIBUTION LIST

	Copies
Chief of Naval Material U. S. Naval Ordnance Laboratory White Oak, Silver Spring, Maryland 20910 Attn: ASW-900 (A. Solem)	1
Commanding Orficer Naval Weapons Center Corona Laboratories Corona, California 91720	1

UNCLASSIFIED

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY (Corporate author) U. S. Naval Ordnance Laboratory White Oak, Silver Spring, Maryland 20910	2a. REPORT SECURITY CLASSIFICATION Unclassified 2b. GROUP

3. REPORT TITLE

Effects of Surface Reflections on Shock Wave Impulse

4. DESCRIPTIVE NOTES (Type of report and inclusive dates)

5. AUTHOR(S) (First name, middle initial, last name)

Verna K. Shuler

6. REPORT DATE 26 October 1968	7a. TOTAL NO. OF PAGES 18	7b. NO. OF REFS 3
8a. CONTRACT OR GRANT NO. b. PROJECT NO. ORDTASK No. ORD-332-001/UF17-354-304 c. d.	9a. ORIGINATOR'S REPORT NUMBER(S) NOLTR 68-138 9b. OTHER REPORT NO(S) (Any other numbers that may be assigned this report)	

10. DISTRIBUTION STATEMENT

Each transmittal of this document outside the agencies of the U. S. Government must have prior approval of NOL.

11. SUPPLEMENTARY NOTES	12. SPONSORING MILITARY ACTIVITY Naval Ordnance Systems Command

13. ABSTRACT

The effects of "surface cut-off" on shock wave impulse from an underwater explosion are discussed. An equation for shock wave "partial impulse," which takes into account the effects of surface cut-off, is discussed and a computer program that generates contours of constant partial impulse is described.

DD FORM 1 NOV 65 1473 (PAGE 1)

S/N 0101-807-6801

UNCLASSIFIED

Security Classification

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
Explosions						
Underwater Explosions						
Underwater Shock Wave						
Underwater Shock Wave Impulse						
Surface Reflection						